

## An Introduction to Scheme

This handout is an introduction to programming in the Scheme programming language. Scheme is a block-structured, lexically-scoped, properly tail-recursive dialect of Lisp that supports first-class functions and continuations. (Don't worry if you don't know what all these terms mean yet – you will by the end of the course.) It was invented by Gerald Jay Sussman and Guy Steele, Jr. at MIT in 1975, and has gained popularity as a teaching language and as an extension language (a language for expressing abstractions built on top of a given set of primitives).

### 1 Scheme = Kernel + Syntactic Sugar + Standard Library

The Scheme language is constructed out of three parts:

1. A small kernel language.
2. A collection of derived forms (a.k.a. **syntactic sugar**) rewritable into kernel constructs.
3. A library of standard procedures. (Unlike in other languages (such as Pascal), we will use the terms “procedure” and “function” interchangeably in the context of Scheme.)

For the most part, we will focus on the kernel language, but will mention syntactic sugar and library procedures as needed. For full details of the language, see the *Revised<sup>5</sup> Report on Scheme (R5RS)*, which is linked from the CS251 home page.

### 2 Syntax of the Scheme Kernel

The **syntax** of a programming language describes the *form* of phrases in the language. In contrast, the **semantics** of the language describes the meaning of these phrases. We describe the syntax of Scheme in this section and the semantics of Scheme in the next section.

The syntax of a language is often specified by a **grammar**. A grammar consists of

- a set of **terminals**, the primitive tokens out of which phrases are constructed.
- a set of **nonterminals**, each of which ranges over a class of compound phrases.
- a set of **productions** that specify the form of the phrases for each nonterminal.

#### 2.1 Terminals

In the case of Scheme, the terminal tokens are as follows:

- $N \in \text{Numbers}$ : e.g., 3, -17, 3.141 (floating point numbers), 6.023e23 (scientific notation), 7/3 (rational numbers), 3+5i (complex numbers);
- $B \in \text{Booleans}$ : #t (true) and #f (false);
- $Y \in \text{Symbols}$ : 'a, 'captain, 'fib\_n-2;

- $C \in \text{Characters}$ : e.g. `#\a`, `#\b`, `#\c`, ..., `#\space`, `#\tab`, `#\newline`;
- $R \in \text{Strings}$ : e.g., "A sample string", "x", ""
- $I \in \text{Identifiers}$ : e.g., `x`, `square`, `fact_n-2`, `$+*/=%.^&!?`. (unlike most languages, Scheme allows many punctuation characters in identifier names);
- reserved words: `lambda`, `if`, `quote`, `letrec`, `begin`, `set!`, `define`, `let`, `cond`, `and`, `or`, `delay`, `cons-stream`;
- parentheses: `(`, `)`.

We use the metavariables  $N$ ,  $B$ ,  $Y$ ,  $C$ ,  $R$ , and  $I$  to range over numbers, booleans, symbols, characters, strings, and identifiers, respectively. For example, we use  $N$  in contexts where we wish to refer to “any number” rather than a particular number.

## 2.2 Nonterminals

Nonterminals specify phrases that are essentially trees whose leaves are terminals. Nonterminals are designated by a metavariable that ranges over a class of phrases. The nonterminals for Scheme are:

- $L \in \text{Literals}$ : “self-evaluating” phrases;
- $E \in \text{Expressions}$ : phrases that denote a value;
- $D \in \text{Definitions}$ : phrases that associate a name with a value;
- $F \in \text{Top-Level Forms}$ : phrases understood by the interpreter; Scheme programs are sequences of these phrases.

## 2.3 Productions

The form of the phrases for each nonterminal are specified by productions that describe the tree structure of the phrases. Each production can be viewed as a rewrite rule. Any phrase consisting only of terminals that can be derived by a sequence of rewrite steps from a given nonterminal is in the syntactic class of that nonterminal.

The productions for the Scheme nonterminals are as follows:

$$\begin{array}{ll}
 L \rightarrow N & \\
 L \rightarrow B & \\
 L \rightarrow Y & \\
 L \rightarrow C & \\
 L \rightarrow R & \\
 \\
 E \rightarrow L & ; \text{Literal} \\
 E \rightarrow I & ; \text{Variable Reference} \\
 E \rightarrow (\text{lambda } (I_{\text{formal}_1} \dots I_{\text{formal}_n}) E_{\text{body}}) & ; \text{Abstraction} \\
 E \rightarrow (E_{\text{rator}} E_{\text{rand}_1} \dots E_{\text{rand}_n}) & ; \text{Application (Call)} \\
 E \rightarrow (\text{if } E_{\text{test}} E_{\text{then}} E_{\text{else}}) & ; \text{Conditional} \\
 \\
 D \rightarrow (\text{define } I_{\text{name}} E_{\text{defn}}) & \\
 \\
 F \rightarrow E & \\
 F \rightarrow D &
 \end{array}$$

Note: for simplicity, a few of the productions for expressions have been omitted above. We will meet them later in the term.

## 2.4 Notes on Scheme Syntax

### 2.4.1 Terminology

Expressions that are terminals (literals and identifiers) are **primitive expressions**. Expressions that are not terminals are **compound expressions**. In order to be able to talk about compound expressions, it is important to know their names and the names of their subparts.

- A compound expression beginning with the reserved word `lambda`, as in `(lambda (I1 ... In) E)`, is an **abstraction**. The identifiers  $I_1 \dots I_n$  are the **formal parameters** (or just **formals**), and  $E$  is the **body**.
- A compound expression that does not begin with a reserved word, as in `(E0 E1 ... En)`, is an **application** (which is synonymous with **procedure call**, **procedure invocation**, **call**, and **invocation**).  $E_0$  is the **operator** (or just **rator**) and  $E_1 \dots E_n$  are the **operands** (or just **rands**).
- A compound expression beginning with the reserved word `if`, such as `(if E1 E2 E3)`, is a **conditional**.  $E_1$  is the **test**,  $E_2$  is the **consequent** (or **then expression**), and  $E_3$  is the **alternative** (or **else expression**).

A compound expression beginning with a reserved word is called a **special form**. For example, conditionals and abstractions are special forms. Applications are the only compound expressions that are not special forms. If you look at the list of reserved words in Section 2.1, you will see that certain special forms have not yet been introduced. Some will be introduced later in this document; others will be introduced later in the term.

## 2.4.2 Parenthesized Expressions Stand for Trees

We will see later in the term that a compound phrases in any language can be viewed as a tree (a so-called **abstract syntax tree (AST)**) whose leaves are terminals.<sup>1</sup> The parenthesized form of compound expressions in Scheme emphasizes the tree-like nature of compound expressions. A parenthesized expression with  $n$  subexpressions represents a tree node with  $n$  subtrees. For example, the expression `(lambda (a) (if (< a 17) #t #\b))` is a sequence of characters that represents the following tree:

You should get into the habit of thinking of expressions as trees. We'll make heavy use of this notion later in the course.

Note that parentheses in Scheme are essential for indicating the tree structure of an expression. Unlike in most other languages, where parentheses are often an optional means of specifying grouping and precedence, parentheses in Scheme are never optional; each one has a very particular meaning!

## 2.4.3 Definitions

It is worth emphasizing that definitions are *not* expressions and cannot be used in the subexpression positions of a compound expression. For example, the following phrase is not legal in Scheme: `(if (< a 17) (define b 1) (define c 2))`. Definitions can only appear at “top-level” – i.e., submitted directly to the interpreter or included as one of the top-level phrases of a file.<sup>2</sup>

## 2.4.4 Strings vs. Symbols

Both strings and symbols are forms of alphabetic data. A string literal is delimited by double quotes, and may contain any characters, including whitespace characters (spaces, tabs, newlines). A symbol literal, on the other hand, is introduced by a single quote but does not have a corresponding closing quote; it consists of a sequence of non-whitespace characters.

From the perspective of abstract data types, strings are more versatile than symbols because strings support numerous operations for manipulating the component characters of a string. On the other hand, symbols are treated as atomic (indivisible) entities; the only interesting thing you can do with a symbol is test it for equality with another symbol. Nevertheless, later on we will see that symbols are very convenient for specifying compound data structures.

Note: if  $I$  is an identifier, then `'I` and `(quote I)` are synonymous. Quotation plays an important role in the specification of compound data structures in Scheme; see Section ?? for details.

---

<sup>1</sup>What makes the tree “abstract” is that it does not include some of the concrete terminals, such as parentheses.

<sup>2</sup>We shall see later that Scheme's block structure allows definitions to also appear in the body of an abstraction, but we'll ignore this for now.

### 2.4.5 Case Sensitivity

Scheme is not case sensitive. This means that the identifiers `fact`, `Fact`, and `FACT` are all treated identically. The same holds true for symbols.

### 2.4.6 Comments

Comments are introduced in Scheme with a semicolon; the comment continues to the end of the line. In MIT Scheme, multi-line comments are opened with `#|` and are closed with `|#`. These are useful for commenting out blocks of code. Multi-line comments nest properly, unlike `/*` and `*/` in C and Java, which do *not* nest properly.<sup>3</sup>

## 3 The Semantics of the Scheme Kernel

In this section, we explore the *meaning* of each of Scheme's syntactic phrases.

### 3.1 The Read-Eval-Print Loop

Scheme is an interpreted language, which means that its syntactic phrases can be directly executed by a virtual machine known as a **Scheme interpreter**. The interpreted nature of Scheme makes it straightforward to evaluate Scheme expressions and name Scheme values in an interactive manner. The entity with which you interact is known as a **read-eval-print loop**, or **REPL** for short.

In order to understand certain details about Scheme evaluation, it is important to have a good model of the REPL. Below is a diagram that depicts a Scheme REPL:

The user interacts with the REPL by typing in a string of characters that represents an expression. The system responds with a sequence of characters that represents the value that results from evaluating the given expression. So all the user sees when interacting with the REPL is character strings. We will represent such interactions as follows:

```
scheme> (* 3 4)
12
```

---

<sup>3</sup>It is my personal opinion that designers of non-nesting multi-line comments should be shot after suitable torture and public humiliation.

The top line contains a prompt (`scheme>`) that designates the name of the language we are interacting with. (This is helpful for distinguishing between the various languages we will be studying.) The prompt is followed by an expression to be evaluated, in this case `(* 3 4)`. The line following the end of the expression is the result printed by the interpreter, in this case `12`.

In order to have a firm understanding of this sort of interaction, it is important to have some idea of what is going on “under the hood”. There are three phases to the REPL:

1. *Read*: In this phase, the character string representation of the expression is **parsed** into an abstract syntax tree.
2. *Eval*: In this phase, the tree representation of the expression is **evaluated** according to the evaluation rules presented later in this section.
3. *Print*: In this phase, the value resulting from the Eval phase is **unparsed** into a character string representation.

The first two phases are fairly uniform among Scheme implementations, but Scheme implementations sometimes differ in how they print out values. For example, some Scheme systems we will use this semester print the symbol `red` as `red` (without a quotation mark) and some will print it as `'red` (with a quotation mark). This is potentially rather confusing, but it helps to realize that the issue is just one of printed representation and not something more fundamental!

In order to evaluate most expressions, the Eval phase must know the meaning of global variable names, which are stored in a table called the **global environment**. When a REPL is first created, this table is initialized with the standard bindings for Scheme: `*` is initially bound to the primitive multiplication procedure, `<=` with the primitive less-than-or-equal-to procedure, and so on. Whenever the user executes a definition – i.e., a phrase of the form `(define I E)` – the REPL evaluates `E` to a value `V`, and then associates the name `I` with `V` in the global environment. This way, the name is available in the evaluation of subsequent expressions.

### 3.2 Evaluation Rules (Substitution Model)

The semantics of Scheme is embodied in the Eval phase of the REPL. This phase determines the value of every expression according to a set of evaluation rules. This semester, we will study several different formulations of these evaluation rules. The formulation presented here, known the **substitution model**, is based on a notion of rewriting an expression according to a set of rewrite rules until it becomes a value. In this context, a value is defined as follows:

A **value** is an expression that is either:

- a literal;
- an abstraction; or
- a variable reference that is the name of a Scheme primitive procedure (e.g. `*`, `<=`).

Below, the evaluation rules are defined by case analysis on the type of expression being evaluated. We shall give examples of each rule using the `subst>` prompt to emphasize that we are using the substitution model. The `subst>` prompt will be followed by the entire sequence of rewritten expressions that lead to the final value.

### 3.2.1 Literals

A literal expression is self-evaluating, so it need not be rewritten: it is already a value.

```
subst> 17
17

subst> #t
#t

subst> #f
#f
; Some versions of Scheme (e.g. MIT Scheme) do not distinguish
; falsity and the empty list. In such versions, #f prints as ().

subst> 'a
'a ; In most Scheme interpreters, this prints as a, not 'a.

subst> #\c
#\c

subst> "Hi there!"
"Hi there!"
```

### 3.2.2 Abstractions

Since abstraction expressions are values, they are self-evaluating:

```
subst> (lambda (x) (* x x))
(lambda (x) (* x x))
; Scheme interpreters differ greatly in terms of how they print
; procedure values. They usually do not print lambda notation. In MIT
; Scheme, user-defined procedures print as #[compound-procedure ZZZ],
; where ZZZ is an identifying number. System-defined procedures (e.g.,
; primitives such as +) have different printed representation

subst> (lambda (a b) (/ (+ a b) 2))
(lambda (a b) (/ (+ a b) 2))

subst> (lambda () 17)
(lambda () 17)
```

### 3.2.3 Variable References

A variable reference expression  $I$  is evaluated as follows:

1. If  $I$  is the name of a standard Scheme primitive procedure, it is self evaluating.

2. If  $I$  is the name of a user-defined value, it rewrites to the most recent value defined by the user.
3. If  $I$  is neither the name of a Scheme primitive procedure nor a user-defined value, the rewriting process halts with an error.

For example, suppose that the following session occurs after invoking a fresh REPL:

*; In a new REPL, the only names bound in the global environment  
; are the standard primitive procedure names of Scheme.*

```
subst> +
+
```

```
; Extend the global environment with a new binding.
subst> (define double (lambda (x) (* x 2)))
double ; The "value" printed after a define is the defined name.
```

```
subst> double
(lambda (x) (* x 2))
```

```
subst> triple
Error! Unbound variable triple
```

### 3.2.4 Conditionals

A conditional expression (`if  $E_1$   $E_2$   $E_3$` ) is evaluated as follows:

1. Rewrite the test expression  $E_1$  into a value  $V_1$ .
2. If  $V_1$  is not `#f`, then the conditional expression rewrites to the then expression  $E_2$ .
3. If  $V_1$  is `#f`, then the conditional expression rewrites to the else expression  $E_3$ .

```
subst> (if (< 1 2) (+ 3 4) (* 5 6))
(if #t (+ 3 4) (* 5 6))
(+ 3 4)
7
```

```
subst> (if (> 1 2) (+ 3 4) (* 5 6))
(if #f (+ 3 4) (* 5 6))
(* 5 6)
30
```

```
subst> (if (< 1 2) (+ 3 4) (5 * 6))
(if #t (+ 3 4) (5 * 6))
(+ 3 4)
7 The error is never discovered in the branch not taken.
```

```
subst> (if 1 2 3)
2
; Any non-false value counts as true in a conditional test
```



### 3.2.5 Applications

An application expression  $(E_0 E_1 \dots E_n)$  is evaluated as a two stage process:

1. In the **subexpression evaluation** stage, all subexpressions  $E_i$  are evaluated to their values  $V_i$ . In this stage, the operator subexpression is not treated any differently than the operand subexpressions. The fact that the operator position is an expression that is evaluated like any other is a source of great power in Scheme that we will witness when we study higher-order functions.
2. In the **application stage**, the operator value  $V_0$  is applied to the operand values  $V_i$  through  $V_n$ . The meaning of application depends on the what kind of value  $V_0$  is:
  - If  $V_0$  denotes a primitive Scheme procedure  $P$ , and there are an appropriate number of arguments of the appropriate type for  $P$ , the application is rewritten to the result of applying  $P$  to the arguments.
  - If  $V_0$  is an abstraction, and the number of arguments matches the number of formal parameters, the application rewrites the application is rewritten to a copy of the body of the abstraction in which the argument values have been substituted for the corresponding formal parameters. (There are some tricky aspects to substitution that are described later.) The value of the application is determined by evaluating the copied body according to the substitution model.
  - In all other cases (i.e.,  $V_0$  is not a procedure, or the number or types of the arguments is not appropriate), the rewriting process halts with an error.

For example:

```
subst> (+ 1 2)
3
```

```
subst> (* (+ 1 2) (- 3 4))
(* 3 -1)
-3
```

```
subst> (not (< 3 2))
(not ())
#t
```

```
subst> (+ 1 (< 2 3))
(+ 1 #t)
Error! Can't add a boolean to a number.
```

```
subst> (not #t #f)
Error! not expects exactly 1 argument, but was given 2.
```

```
subst> (2 + 3)
Error! 2 is not a procedure.
```

```
subst> ((lambda (x) (* x x)) (+ 2 3))
((lambda (x) (* x x)) 5) ; No substitution until arg is value
(* 5 5)
25
```

```
subst> (define average (lambda (a b) (/ (+ a b) 2)))
average
```

```
subst> (average (+ 2 4) (* (+ 1 2) 4))
((lambda (a b) (/ (+ a b) 2)) 6 (* 3 4))
((lambda (a b) (/ (+ a b) 2)) 6 12) ; No substitution until
(/ (+ 6 12) 2) ; both args are values
(/ 18 2)
9
```

```
subst> (define sum-of-squares ; This method of formatting code
      (lambda (x y) ; is called pretty-printing.
        (+ (square x) ; Note that it's OK that square
            (square y)))) ; is not yet defined
sum-of-squares
```

```
subst> (define square (lambda (x) (* x x)))
square
```

```
subst> (sum-of-squares (+ 1 2) (/ 12 3))
((lambda (x y) (+ (square x) (square y))) 3 4)
(+ (square 3) (square 4))
(+ ((lambda (x) (* x x)) 3) ((lambda (x) (* x x)) 4))
(+ (* 3 3) (* 4 4))
(+ 9 16)
25
```

```
subst> (define add-a (lambda (x) (+ a x)))
add-a
```

```
subst> (define a 1)
a
```

```
subst> (add-a 3)
((lambda (x) (+ a x)) 3)
(+ a 3)
(+ 1 3)
4
```

```
subst> (define a 17) ; Overwrites previous definition of a

subst> (add-a 3)
((lambda (x) (+ a x)) 3)
(+ a 3)
(+ 17 3)
20

subst> ((+ 1 2) (* 3 4) (- 5 6))
(3 12 -1)
Error! 3 is not a procedure.

subst> ((lambda (x) (x x)) (lambda (x) (x x)))
((lambda (x) (x x)) (lambda (x) (x x)))
((lambda (x) (x x)) (lambda (x) (x x)))
((lambda (x) (x x)) (lambda (x) (x x)))
... ; This shows that some rewriting processes may not terminate
```

### 3.2.6 Substitution

The substitution model gets its name from the substitution step that takes place in the application of an abstraction to argument values. Although substitution seems fairly intuitive, there are some pitfalls to watch out for:

- In cases where the same identifier names logically distinct variables, we must be careful to only substitute for the appropriate variable references. For example, in the expression,

```
((lambda (a) (if (< 3 a)
                (lambda (b) (* a b))
                (lambda (a) (* a a))))
2)
```

there are really two distinct variables named `a`: the one declared by the outer `lambda` and the one declared by the inner `lambda`. Subscripting the formal parameters and their associated variable references would yield:

```
((lambda (a1) (if (< 3 a1)
                (lambda (b1) (* a1 b1))
                (lambda (a2) (* a2 a2))))
2)
```

The result of substitution should therefore be:

```
(if (< 3 2) (lambda (b) (* 2 b)) (lambda (a) (* a a)))
```

That is, we should not substitute 2 for the `as` in `(lambda (a) (* a a))` because they stand for a different variable than the one we are substituting for. We shall formalize this notion when we study the scope of names in programs.

- We must be careful that Scheme primitive procedures passed as arguments are not accidentally capture by formal parameters that happen to have the same name. For example, in the following expression

```
((lambda (f) (lambda (+) f)) +)
```

a naive substitution yields an identity function, which is incorrect:

```
(lambda (+) +)
```

The correct result is a procedure that always ignores its argument and returns +:

```
(lambda (+_1) +)
```

This result can be achieved by consistently renaming the formal parameter (and any associated variable references) of an abstraction when substituting a value into the abstraction that might accidentally be “captured”. Again, we shall formalize this notion when we study the scope of names in programs.

### 3.3 Substitution Model Interpreter

Normally, a Scheme REPL only shows you the final result of evaluating an expression. It does not show you the intermediate sequence of rewritten expressions.

I have developed an alternative Scheme REPL that displays all the intermediate rewritten expressions as implied by the substitution model. I shall refer to this REPL as the substitution model REPL. The substitution model REPL is a pedagogical tool that will help you to understand the details of the substitution model I encourage you to use the substitution model REPL at the beginning of the course whenever you have questions about the evaluation of an expression.

You are strongly encouraged to experiment with the substitution model interpreter, which is located in the CVS-controlled directory `cs251/subst-model`. Once you have created your local `cd251` CVS repository (see a later handout for details), you can use the substitution model interpreter as follows:

1. Evaluate `(cd "~/cs251/subst-model")` in the Scheme REPL. This changes the default file directory to be the directory containing the implementation of the substitution model REPL.
2. Evaluate `(load "subst-model.scm")` in the Scheme REPL. This loads the definitions that implement the substitution model REPL.
3. Evaluate `(subst-repl)` in the Scheme REPL. This will enter a new REPL whose prompt is `subst>` (to distinguish it from the normal Scheme REPL).

Here are some usage notes for the substitution model REPL:

- As in normal Scheme, you can define values and load files containing definitions and expressions.
- If you evaluate an expression that signals an error, the error will exit the substitution model REPL and return to the normal Scheme REPL. To restart the substitution model REPL, evaluate `(subst-repl)` again. Unfortunately, all definitions from your previous session will have been lost. For this reason, it is a good idea to keep all your definitions in a file, so it is easy to reload them via `load`.

- To exit the substitution model REPL, evaluate the special token `exit`.

The substitution model REPL implements a design decision that is worth explaining. Where possible, it performs rewrite steps on the subexpressions of an application in parallel. For example, consider the first step the average example from above

```
subst> (average (+ 2 4) (* (+ 1 2) 4))
((lambda (a b) (/ (+ a b) 2)) 6 (* 3 4))
```

In one “step”, the interpreter has rewritten `average` to its associated value, `(+ 2 4)` to `6`, and `(* (+ 1 2) 4)` to `(* 3 4)`. The substitution model allows, but does not require, all of these rewrites to occur in the same step. There are many alternative strategies. For example, they could be performed one-by-one in a left-to-right manner:

```
subst> (average (+ 2 4) (* (+ 1 2) 4))
((lambda (a b) (/ (+ a b) 2)) (+ 2 4) (* (+ 1 2) 4)) ; Rewrite average
((lambda (a b) (/ (+ a b) 2)) 6 (* (+ 1 2) 4))      ; Rewrite (+ 2 4)
((lambda (a b) (/ (+ a b) 2)) 6 (* 3 4))           ; Rewrite (+ 1 2)
((lambda (a b) (/ (+ a b) 2)) 6 12)                ; Rewrite (* 3 4)
```

Another approach would be a right-to-left order:

```
subst> (average (+ 2 4) (* (+ 1 2) 4))
(average (+ 2 4) (* (+ 1 2) (* 3 4))) ; Rewrite (+ 1 2)
(average (+ 2 4) 12)                  ; Rewrite (* 3 4)
(average 6 12)                         ; Rewrite (+ 2 4)
((lambda (a b) (/ (+ a b) 2)) 6 12)   ; Rewrite average
```

Many other orders of evaluation are possible. It turns out that the particular order taken cannot affect the final result in the purely functional subset of Scheme (i.e., the subset in which no side-effects are allowed).

### 3.4 Procedure Tracing

The detailed step-by-step information provided by the substitution model interpreter can be overwhelming for all but the simplest examples. Sometimes you just want to focus on the invocation of a small number of procedures. The procedure tracing facility of MIT Scheme is extremely handy for this purpose. It allows you to select any number of procedures you wish to “trace”, and then prints out information about these procedures each time they are called or return. See Section 4.3 of the MIT Scheme handout (#4) or Section 5.4 of the MIT Scheme User’s Manual for more information.

### 3.5 Recursion and Iteration

It is straightforward to express recursive definitions in Scheme. For example, here is a recursive definition in Scheme of the classic factorial function:

```
(define fact-rec
  (lambda (n)
    (if (= n 0)
        1
        (* n (fact-rec (- n 1))))))
```

The substitution model explains the evaluation of calls to `fact-rec` without a hitch. For example:

```
subst> (fact-rec 3)
(fact-rec 3)
((lambda (n) (if (= n 0) 1 (* n (fact-rec (- n 1))))) 3)
(if (= 3 0) 1 (* 3 (fact-rec (- 3 1))))
(if #f 1 (* 3 (fact-rec (- 3 1))))
(* 3 (fact-rec (- 3 1)))
(* 3 ((lambda (n) (if (= n 0) 1 (* n (fact-rec (- n 1))))) 2))
(* 3 (if (= 2 0) 1 (* 2 (fact-rec (- 2 1)))))
(* 3 (if #f 1 (* 2 (fact-rec (- 2 1)))))
(* 3 (* 2 (fact-rec (- 2 1))))
(* 3 (* 2 ((lambda (n) (if (= n 0) 1 (* n (fact-rec (- n 1))))) 1)))
(* 3 (* 2 (if (= 1 0) 1 (* 1 (fact-rec (- 1 1)))))
(* 3 (* 2 (if #f 1 (* 1 (fact-rec (- 1 1)))))
(* 3 (* 2 (* 1 (fact-rec (- 1 1)))))
(* 3 (* 2 (* 1 ((lambda (n) (if (= n 0) 1 (* n (fact-rec (- n 1))))) 0))))
(* 3 (* 2 (* 1 (if (= 0 0) 1 (* 0 (fact-rec (- 0 1)))))
(* 3 (* 2 (* 1 (if #t 1 (* 0 (fact-rec (- 0 1)))))
(* 3 (* 2 (* 1 1)))
(* 3 (* 2 1))
(* 3 2)
6
```

Note that there is no need to introduce notions of function activations or stacks, nor is there any need to handle recursive functions applications any differently than non-recursive ones. The same rules that handle “regular” function applications handle recursive ones.

It’s worth noting that a sort of stack-like structure natural emerges out of the rules for evaluating an application. Recall that all subexpressions of an application must be rewritten to a value before the operator can be applied to the resulting values. This means that operator applications remain “pending” while their arguments are being calculated; this gives rise to what is normally known as the call stack. This stack-like nature of recursive procedure call evaluation is more apparent in a more stylized rewrite sequence that only highlights certain intermediate expressions:

```
subst> (fact-rec 5)
(fact-rec 5)
(* 5 (fact-rec 4))
(* 5 (* 4 (fact-rec 3)))
(* 5 (* 4 (* 3 (fact-rec 2))))
(* 5 (* 4 (* 3 (* 2 (fact-rec 1)))))
(* 5 (* 4 (* 3 (* 2 (* 1 (fact-rec 0)))))
(* 5 (* 4 (* 3 (* 2 (* 1 1)))))
(* 5 (* 4 (* 3 (* 2 1)))
(* 5 (* 4 (* 3 2)))
(* 5 (* 4 6))
(* 5 24)
120
```

The growing and shrinking “bulge” above is characteristic of the stack-like nature of a recursive process.

Of course, it is possible to express tree recursions as well as linear recursions. Here is the Scheme version of the classic Fibonacci function

```
(define fib-rec
  (lambda (n)
    (if (< n 2)
        n
        (+ (fib-rec (- n 1)) (fib-rec (- n 2))))))
```

and here is a manually formatted transcript of the call (fib-rec 4):

```
subst> (fib-rec 4)
(+ (fib-rec 3) (fib-rec 2))
(+ (+ (fib-rec 2) (fib-rec 1))
  (+ (fib-rec 1) (fib-rec 0)))
(+ (+ (+ (fib-rec 1) (fib-rec 0))
      1)
  (+ 1 0))
(+ (+ (+ 1 0)
      1)
  1)
(+ (+ 1
      1)
  1)
(+ 2
  1)
3
```

What about iteration? Since Scheme does not provide any mutable variables or iterative constructs like `while`, `for`, or `repeat/until`, is it possible to express iteration in Scheme? For example, an iterative solution to computing the factorial of 5 would be to update two state variables in a loop as follows:

Number	Result
5	1
4	5
3	20
2	60
1	120
0	120

It turns out that it is possible to express this sort of iterative process in Scheme without mutable variables or iteration constructs. All that is needed is procedure calls! The iterative factorial process is expressed in Scheme using two procedures:

```

(define fact-iter
  (lambda (n)
    (fact-loop n 1)))

(define fact-loop
  (lambda (number result)
    (if (= number 0)
        result
        (fact-loop (- number 1) (* result number)))))

```

The `fact-iter` procedure is a one-argument procedure that is the entry point for starting the process. All it does is to call `fact-loop`, a procedure whose two arguments represent the state variables in the above table, with the appropriate initial values. Here is a stylized trace of the substitution model on a sample call:

```

subst> (fact-iter 5)
(fact-loop 5 1)
(fact-loop 4 5)
(fact-loop 3 20)
(fact-loop 2 60)
(fact-loop 1 120)
(fact-loop 0 120)
120

```

Each line corresponds precisely to one row of the above table. Note how Scheme avoids the need for mutable variables in this example. Rather than creating two state variables and changing their contents over time, each call to `fact-loop` effectively introduces new copies of the `number` and `result` parameters. Scheme avoids the need for iteration constructs by simply using an explicit procedure call to reenter the loop. Note that this process does not have a growing and shrinking bulge but takes a constant amount of space throughout. The reason is that there are no pending operations that are waiting for `fact-loop` to return. A procedure call that is not nested within pending operations is said to be a tail call. Scheme implementations are required to be properly tail recursive, which means that they must implement tail calls without pushing any information on the run-time stack of execution frames. This guarantees that processes like those generated by `fact-loop` will exhibit the same desirable constant space behavior as `while` and `for` loops in Java, C, and other languages with special looping constructs.

## 4 Scheme Syntactic Sugar

Syntactic sugar specifies new special forms via rewrites into the kernel language. This allows the language to be extended with new special forms without affecting its essential core. Below are some of the most important desugarings; for more, see Section 7.3 of *R5RS*. Note that it is possible to write any Scheme program without using any syntactic sugar at all. However, syntactic sugar helps to makes programs more concise and more readable.



```

(define (Iname Iformal_1 ... Iformal_n) Ebody)
  ~> (define Iname (lambda (Iformal_1 ... Iformal_n) Ebody))

(let ((Iname_1 Edef_1) ... (Iname_n Edef_n)) Ebody)
  ~> ((lambda (Iname_1 ... Iname_n) Ebody) Edef_1 ... Edef_n)

(cond) ~> unspecified
(cond (else E)) ~> E
(cond (Etest Eaction) ...)
  ~> (if Etest Eaction (cond ...))

(and) ~> #t
(and E) ~> E
(and E1 E2 ...) ~> (if E1 (and E2 ...) #f)

(or) ~> #f
(or E) ~> E
(or E1 E2 ...) ~> (let ((I E1)) (if I I (or E2 ...))),
  where I is a “fresh” identifier (i.e., not mentioned elsewhere in the program.)

```

The above desugaring rules make it possible to write many definitions and expressions much more concisely than would be possible in kernel Scheme. For example:

```

(define (f x y z)
  (cond ((and (< x y) (< x z)) (* x y z))
        ((or (> x z) (> y z)) (+ x y z))
        (else (let ((a (+ x y))
                    (b (- y z)))
                (/ (* a a) (* b b))))))

```

is the “sugared” form of the following kernel Scheme definition:

```

(define f
  (lambda (x y z)
    (if (if (< x y)
           (< x z)
           #f)
        (* x y z)
        (if ((lambda (.temp.)
              (if (.temp.
                  .temp.
                  (> y z)))
            (> x z))
            (+ x y z)
            ((lambda (a b)
              (/ (* a a) (* b b)))
             (+ x y)
             (- y z))))))

```

## 5 The Scheme Standard Library

The Scheme grammar defined in Section 2 is extremely small and simple. In fact, it seems too small. Is it really enough to support the writing of arbitrarily complex programs? The answer is yes – as long as these kernel language constructs are used in conjunction with a suitable library of standard functions and some convenient syntactic sugar.

Scheme comes equipped with a number of datatypes and operations on those datatypes. One way to extend the language is to add a new datatype with its associated operations; for these kinds of extensions the kernel remains unmodified. Following are some of the standard datatypes and their common operations; for more examples, see Section 6 of *R5RS*.

- *Numbers*: +, -, \*, /, max, min, quotient, remainder, <, <=, =, >, >=, ...
- *Booleans*: not
- *Symbols*: eq?, symbol->string, string->symbol
- *Characters*: char<?, char=?, char>?, char->integer, integer->char
- *Strings*: string<?, string=?, string-ref, ...
- *Procedures*: apply
- *Pairs*: cons, car, cdr, ...
- *Lists*: list, list-ref, list-tail, append, reverse, map, ...
- *Vectors*: vector, vector-ref, ...

For each standard datatype, there is a predicate that tests to see if an element is of that datatype – e.g., `number?`, `boolean?`, `symbol?`, `string?`, `procedure?`, `pair?`, `list?`, `null?`, `vector?`, ...

Most datatypes are equipped with an equality predicate that tests whether two elements of the datatype are the same. Here are the standard equality predicates for the atomic (i.e., not compound) datatypes:

- *Numbers*: =
- *Booleans*: eq?
- *Symbols*: eq?
- *Characters*: char=?
- *Strings*: string=?

Rather than remember the above, you can just use the `eqv?` procedure, which dispatches to the appropriate equality testing procedure depending on the type of its argument.

Equality testing is trickier for compound data structures (pairs, lists, vectors) and procedures. For now, we will assume that two compound data structures are the same if they have the same shape and all their corresponding atomic components are pairwise `eqv?` This sort of equality can be tested with the `equal?` procedure.

As for procedures, what does equality on procedures mean? Ideally we would like what is called an “extensional equality”: two procedures are equal if they return the same results on all inputs.

Unfortunately, such an equality test is uncomputable; it is a prime example of a mathematical function that is well-defined, but for which it is impossible to write a program. A much easier equality to test is whether two procedures were created with the evaluation of the same `lambda` expression at the same point in time; this can be tested with `eq?`.